
Internet Of Things
Workshop
Batteriebetriebener Temperatursensor
ESP8266 (Wemos D1.mini), Arduino IDE, 
WiFi, MQTT, DeepSleep



Agenda
● Hardware bauen, Löten
● Arduino IDE
● MQTT
● WiFi Konfigurationen mit ESP8266 unter der Arduino 

IDE
● DeepSleep
● Berechnungen zur Batterielaufzeit
● Weiterführende Links



Hardware bauen, Löten



Benötigte Materialien
● Wemos D1.mini

● Stiftleisten (sind meistens mitgliefert

● 4.7 kOhm Widerstand

● Temperatursensor DS18B20

● Batteriebox (alternativ auch eine 3er-
Box aus Conrad/Völkner/Reichelt/….

● Nicht abgebildet: Lötkolben, Breadboard 
(zum einfacheren Löten), etwas kleine 
Kabel



Löten: Stiftleisten

● Stiftleisten auf ein Breadboard einstecken
● Darauf die Buchsenleisten stecken
● Darauf den Wemos D1.mini legen
● Die Buchsenleisten liegen nun gerade an
● Löten: 2-3 Sekunden den Stift UND den 

Kupferring mit dem Lötkolben (~320°C) 
erhitzen, dann seitlich den Lötzinn 
zuführen (nicht an den Lötkolben selbst)



Löten: Batterieanschluss

● Den Wemos D1.mini auf dem Breadboard eingesteckt lassen.

● Den Plus-Pol (üblicher Weise rotes Kabel) an die 3V3-Schnittstelle anlöten
● Den Minus-Pol (üblicher Weise schwarzes Kabel) an G (für “Ground”) anlöten



Hardware zusammenstecken

● Drahtbrücke zwischen D0 und RST 
herstellen

● Sonst wacht der ESP8266 aus dem 
DeepSleep nicht mehr auf

● Temperatursensor mit der flachen Seite 
in Richtung Mitte des D1.mini 
einstecken
● D5, D6, D7

● Pull-Up-Widerstand 4.7 kOhm zwischen 
3V3 und D6 (mittlerer 
Temperatursensor-Pin)

● Mit etwas Quetschen hält das gut 
zusammen, die Drähtchen vom 
Temperatursensor sind recht dünn



2 oder 3 AA-Batterien
● 2 Varta Industrial AA Alkaline liefern 3V
● 3 enloop pro AA („Akkus“) liefern 3.6V



Arduino IDE



Arduino IDE
● https://www.arduino.cc
● Ursprünglich für Arduino-Kompatible 

Boards vorgesehen
● Programmiersprache C
● Neue Boards (in diesem Fall 

ESP8266) können eingebunden 
werden

● Datei → Voreinstellungen → 
Zusätzliche Boardverwalter-URL:
http://arduino.esp8266.com/stable/
package_esp8266com_index.json

● Werkzeuge → Board → 
Boardverwalter → esp8266

● Jetzt kann man ESP8266-Boards mit 
der Arduino IDE programmieren

https://www.arduino.cc/
http://arduino.esp8266.com/stable/package_esp8266com_index.json
http://arduino.esp8266.com/stable/package_esp8266com_index.json


Arduino IDE: Minimumprogramm

void setup() {

  // put your setup code here, to run once:

}

void loop() {

  // put your main code here, to run repeatedly:

}



MQTT



MQTT
● Message Queue Telemetry Transport
● Nachrichtenprotokoll für Machine-to-Machine-Kommunikation 

(M2M)
● Typisches Protokoll für “Internet of Things”
● Grundprinzip: Ein Broker “Händler” erhält alle Informationen 

und hält diese für alle Teilnehmer vor, der letzte Wert wird 
zurückgegeben

● Topic-Orientiert (“haus”, “iot-device”, …)
● Hierarchisch (“haus/bad/sensor/temperatur”)
● In der IDE die Library einbinden AsyncMqttClient



MQTT verwenden
● Eigener Server mit quelloffenem Broker 

„mosquitto“
● Alternativ: test.mosquitto.org

– Öffentlich (jeder kann mitlesen)
– Kein garantierter Service
– Basistopic wählen, z.B. eigenen Nickname und/oder 

beliebige Zeichenfolge



WiFi Konfigurationen mit ESP8266 
unter der Arduino IDE



WiFi Konfiguration ESP8266
● Library ESP8266WiFi einbinden

const char* ssid = "CTDO-IoT";

const char* password = "12345678";

WiFi.begin(ssid, password);

while (WiFi.status() != WL_CONNECTED) {

    delay(500);

    Serial.print(".");

  }

// Print the IP address

Serial.println(WiFi.localIP());



DeepSleep



DeepSleep
● ESP8266 kann in einen “Tiefschlaf” 

versetzt werden
● Man muss eine Aufwachzeit

in ms angeben
● Zur Aufwachzeit sendet der ESP8266 

ein Signal an den digitalen Ausgang 
D0 (das ist der GPIO16)

● Wenn D0 mit einer Drahtbrücke an 
RST angeschlossen ist, startet damit 
der ESP8266 neu



DeepSleep
const int sleepTimeS = 295; // 5 Minuten insgesamt

void initiateDeepSleep()

{

  ESP.deepSleep(sleepTimeS * 1000000);

  delay(100); 

}



Berechnungen zur Batterielaufzeit



Berechnungen zur Batterielaufzeit
● Eingeschaltet (WiFi suchen, Paket versenden)
● Typisch ~5 Sekunden lang, ca. 100mA
● Pause 295 Sekunden (5 Minuten), ca. 220µA
● Zykluslänge damit 300 Sekunden, Messungen bei 3V

(zwei VARTA Industrial 1.5V-Alkaline):
● 5s * 100mA = 500mAs
● 295s * 0,22mA = 64,9mAs
● 564,9mAs  je 300 Sekunden = 1,883 mA im Schnitt
● 2600mAh in den Batterien → 1380h = 57 Tage
● Eine Verlängerung der Pause bewirkt längere Laufzeit,

z.B. nur ein Wert pro Stunde: 479 Tage



Weiterführende Links



Weiterführende Links
● Blog-Artikel mit detaillierter Beschreibung dieses 

IoT-Workshops: https://www.anginf.de/?p=526
● MQTT-Broker: https://mosquitto.org/
● WEMOS D1.mini: 

https://wiki.wemos.cc/products:d1:d1_mini
●

●

https://www.anginf.de/?p=526
https://mosquitto.org/
https://wiki.wemos.cc/products:d1:d1_mini


Anhang: Komplettes Programm



Quellcode (nur zur Info)



Quellcode (nur zur Info)



Quellcode (nur zur Info)



Quellcode (nur zur Info)



Quellcode (Kopierbar)
// Wemos D1 board, connected to a battery box and a DS18B20 temperature sensor

//

 

// For temperature reading

// Libraries needed:

// * OneWire

// * DallasTemperature

//

// Pinout: https://wiki.wemos.cc/products:d1:d1_mini

// D0 = GPIO16 --> Connect D0 to RST for Deep Sleep-Wakeup

 

#include <OneWire.h> 

#include <DallasTemperature.h>

 

const char* ssid = "ENTER_YOUR_SSID_HERE";

const char* password = "ENTER_YOUR_WLAN_PASS_HERE";

 

#define DEVICENAME "maintopic/devicename"

#define TOPIC DEVICENAME"/temperature"

#define ONLINETOPIC DEVICENAME"/online"

#define MQTTSERVER IPAddress(37, 187, 106, 16) // test.mosquitto.org = 37.187.106.16

const int sleepTimeS = 300; // Reduce this value for debugging. Increase if you want more battery life

 

#define VCCPIN D7

#define ONE_WIRE_BUS D6

#define GNDPIN D5

 

OneWire oneWire(ONE_WIRE_BUS); 

DallasTemperature sensors(&oneWire);

 

float tempC;

 

// For WLAN & MQTT

#include <ESP8266WiFi.h>

#include <AsyncMqttClient.h>

AsyncMqttClient mqttClient;

uint16_t packetId1Pub;

bool packet1Ack = false;

 

bool ready = false;

 

char *ftoa( double f, char *a, int precision)

{

 long p[] = {0,10,100,1000,10000,100000,1000000,10000000,100000000};

 

 char *ret = a;

 long heiltal = (long)f;

 itoa(heiltal, a, 10);

 while (*a != '\0') a++;

 *a++ = '.';

 long desimal = abs((long)((f - heiltal) * p[precision]));

 itoa(desimal, a, 10);

 return ret;

}

 

void onMqttPublish(uint16_t packetId) {

  Serial.println("** Publish acknowledged **");

  Serial.print("  packetId: ");

  Serial.println(packetId);

  // TODO: Only when both packages were acknowledged...

  if (packetId == packetId1Pub) {

    packet1Ack = true;

  }

  if (packet1Ack) {

    ready = true;

  }

}

 

void onMqttConnect(bool sessionPresent) {

  char buf[7];

  packetId1Pub = mqttClient.publish(TOPIC, 1, true, ftoa(tempC, buf, 2));

}

 

 

void setup() {

  pinMode(GNDPIN, OUTPUT);

  pinMode(VCCPIN, OUTPUT);

  digitalWrite(GNDPIN, LOW);

  digitalWrite(VCCPIN, HIGH);

  Serial.begin(115200); 

  Serial.println("ESP-Temperature-Reader-and-MQTT-Poster-via-WiFi"); 

  // Start up the sensors library 

  sensors.begin(); 

}

 

void loop() {

  // Send the command to get temperature readings 

  Serial.println("Requesting Temperature"); 

  sensors.requestTemperatures();

 

  // You can have more than one DS18B20 on the same bus.  

  // 0 refers to the first IC on the wire 

  Serial.println("Requesting Temperature from Device 0"); 

  tempC = sensors.getTempCByIndex(0);

 

Serial.println("Connecting to WIFI"); 

  // Connect to WiFi

  WiFi.begin(ssid, password);

  int timeout = 0;

  while (WiFi.status() != WL_CONNECTED) {

    timeout++;

    if (timeout>20) {

        // WIFI isn't available after 10 seconds -> abort mission, mission's a failure

        initiateDeepSleep();

      }

    delay(500);

    Serial.print(".");

  }

  Serial.println("");

  Serial.println("WiFi connected");

 

  // Print the IP address

  Serial.println(WiFi.localIP());

 

  // Publish result to MQTT

  mqttClient.onConnect(onMqttConnect);

  mqttClient.onPublish(onMqttPublish);

  mqttClient.setServer(MQTTSERVER, 1883);

  mqttClient.setKeepAlive(5).setCleanSession(false).setWill(ONLINETOPIC, 2, true, "no"); // .setCredentials("user", "pass").setClientId(DEVICENAME);

  Serial.println("Connecting to MQTT...");

  mqttClient.connect();

 

  timeout = 0;

  while (!ready) {

    delay(250);

    timeout++;

    if (timeout > 40)

    {

        // MQTT isn't available after 10 seconds -> abort mission, mission's a failure

        initiateDeepSleep();

    }

    Serial.print(".");

  }

  Serial.println("");

  initiateDeepSleep();

}

 

void initiateDeepSleep()

{

  ESP.deepSleep(sleepTimeS * 1000000);

  delay(100); 

}

